PROC SQL for DATA Step Die-hards
Christianna S. Williams, Chapel Hill, North Carolina

ABSTRACT
PROC SQL can be a bit daunting for the 'old dogs' who have learned SAS data management techniques using exclusively the DATA Step. However, when it comes to data manipulation, SAS often provides more than one method to achieve the same result, and SQL provides another valuable tool to have in one's repertoire. Further, Structured Query Language is implemented in many widely used relational database systems with which SAS may interface, so it is a worthwhile skill to have from that perspective as well.

This paper will present a series of increasingly complex examples and hands-on exercises. In each case I will demonstrate the DATA Step method with which users are probably already familiar, followed by SQL code that will accomplish the same data manipulation. The simplest examples include subsetting variables (columns, in SQL parlance) and observations (rows), while the most complex situations will include MERGEs (JOINS) of several types and the summarization of information over multiple observations for BY groups of interest. The comparative, task-oriented approach is aimed at helping accomplished DATA step programmers "see" what different SQL clauses achieve. Conversely, SQL experts may pick up a few esoteric DATA step tricks. The participant should come away with a solid grounding in Data Step to PROC SQL translation as well as some new SQL "vocabulary" – even die-hard Data Step-ers may find themselves "thinking in SQL"!

INTRODUCTION
The DATA step is a real workhorse for virtually all SAS users. Its power and flexibility are probably among the key reasons why the SAS language has become so widely used by data analysts, data managers and other "IT professionals". However, at least since version 6.06, PROC SQL, which is the SAS implementation of Structured Query Language, has provided another extremely versatile tool in the base SAS arsenal for data manipulation. Still, for many of us who began using SAS prior to the addition of SQL or learned from hardcore DATA step programmers, change may not come easily. We are often too pressed for time in our projects to learn something new or venture from the familiar, even though it may save us time and make us stronger programmers in the long run. Often SQL can accomplish the same data manipulation task with considerably less code than more traditional SAS techniques.

This paper is designed to be a relatively painless introduction to PROC SQL for users who are already quite adept with the DATA step. Several examples of row selection, grouping, sorting, summation and combining information from different data sets will be presented. For each example, I’ll show a DATA step method (recognizing that there are often multiple techniques to achieve the same result) followed by an SQL method. Throughout the paper, when I refer to “DATA step methods”, I include under this term other base SAS procedures that are commonly used for data manipulation (e.g. SORT, SUMMARY). In each code example, SAS keywords are in ALL CAPS, while arbitrary user-provided parameters (i.e. variable and data set names) are in lower case.

THE DATA
First, a brief introduction to the data sets. Table 1 describes the four logically linked data sets, which concern the hospital admissions for twenty make-believe patients. The variable or variables that uniquely identify an observation within each data set are indicated in bold; the data sets are sorted by these keys. Complete listings are included at the end of the paper. Throughout the paper, it is assumed that these data sets are located in a data library referenced by the libref EX.

<table>
<thead>
<tr>
<th>Data set</th>
<th>Variable</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>admissions</td>
<td>pt_id*</td>
<td>patient identifier</td>
</tr>
<tr>
<td></td>
<td>admdate</td>
<td>date of admission</td>
</tr>
<tr>
<td></td>
<td>disdate</td>
<td>date of discharge</td>
</tr>
<tr>
<td></td>
<td>hosp</td>
<td>hospital identifier</td>
</tr>
<tr>
<td></td>
<td>bp_sys</td>
<td>systolic blood pressure (mmHg)</td>
</tr>
<tr>
<td></td>
<td>bp_dia</td>
<td>diastolic blood pressure (mmHg)</td>
</tr>
<tr>
<td></td>
<td>dest</td>
<td>discharge destination</td>
</tr>
<tr>
<td></td>
<td>primdx</td>
<td>primary diagnosis (ICD-9)</td>
</tr>
<tr>
<td></td>
<td>md</td>
<td>admitting physician identifier</td>
</tr>
</tbody>
</table>
### Data set | Variable | Description
--- | --- | ---
patients | id | patient identifier
 | lastname | patient last name
 | firstname | patient first name
 | sex | gender (1=M, 2=F)
 | dob | date of birth
 | primmd | primary physician identifier
 | zipcode | patient residence zip code

| hospitals | hosp_id | hospital identifier
 | hospname | hospital name
 | zip | hospital zip code
 | beds | number of beds
 | has_er | presence of ER (Y/N)

| doctors | md_id | physician identifier
 | hospadm | hospital at which MD has admitting privileges
 | lastname | physician last name
 | zip_office | *bold identifies key fields (needed to uniquely identify a record in each data set); data sets are sorted by these keys.

**EXAMPLE 1: SUBSETTING VARIABLES (COLUMNS)**

In this first, extremely simple example, we just want to select three variables from the ADMISSIONS data set.

**DATA step code:**

```
DATA selvar1 ;
SET ex.admissions (KEEP = pt_id admdate disdate);
RUN;
```

In the DATA step, the KEEP= option on the SET statement does the job.

**SQL code:**

```
PROC SQL;
   CREATE TABLE selvar2 AS
      SELECT pt_id, admdate, disdate
      FROM ex.admissions ;
QUIT;
```

The SQL procedure is invoked with the PROC SQL statement. SQL is an interactive procedure, in which RUN has no meaning. QUIT forces a step boundary, terminating the procedure. An SQL table in SAS is identical to a SAS data set. The output table could also be a permanent SAS data set; in such case, it would be referenced by a two-level name (e.g. EX.SELVAR2). A few other features of this simple statement are worth noting. First, the variable names are separated by commas rather than spaces; this is a general feature of lists in SQL – lists of tables, as we’ll see later are also separated by commas. Second, the AS keyword signals the use of an alias; in this case the table name SELVAR2 is being used as an alias for the results of the query beginning with the SELECT clause. We’ll see other types of aliases later. Third, the FROM clause names what entity we are querying. Here it is a single input data set (EX.ADMISSIONS), but it could also be multiple data sets, a query, a view (either a SAS view or a SAS/ACCESS view), or a table in an external database. Examples of the first two types will be presented below.

SQL can also be used to write reports, in which case the statement above would begin with the SELECT clause. The resulting report looks much like output from PROC PRINT. SAS views, which are stored queries, can also be created with SQL. To do this, the keyword TABLE in the CREATE statement above would simply be replaced with the keyword VIEW. In this paper, since I am focusing on the generation of new data sets meeting desired specifications, virtually all the SQL statements will begin with “CREATE TABLE...”.

---

**Hands-On Workshops**

NESUG 2012
One final point before we move on to some more challenging examples: interestingly, although the results of the DATA step and the PROC SQL are identical (neither PROC PRINT nor PROC COMPARE reveal any differences), slightly different messages are generated in the log.

For the DATA step:
NOTE: The data set WORK.SELVAR1 has 25 observations and 3 variables.

For PROC SQL:
NOTE: Table WORK.SELVAR2 created, with 25 rows and 3 columns.

This demonstrates a distinction in the terminology that stems from the fact that SQL originated in the relational database arena, while, of course, the DATA step evolved for “flat file” data management. Table 2 shows these equivalencies.

Table 2. Equivalent SAS terms between the DATA step and PROC SQL

<table>
<thead>
<tr>
<th>DATA step</th>
<th>PROC SQL</th>
</tr>
</thead>
<tbody>
<tr>
<td>data set</td>
<td>table</td>
</tr>
<tr>
<td>observation</td>
<td>row</td>
</tr>
<tr>
<td>variable</td>
<td>column</td>
</tr>
</tbody>
</table>

EXAMPLE 2: SELECTING OBSERVATIONS (ROWS)

Almost all of the rest of the examples involve the selection of certain observations (or rows) from a table or combinations of tables. Here we simply want to select admissions to the Tarheel hospital (HOSP EQ 3 on the ADMISSIONS data set).

DATA step code:
DATA hosp3_adm1;
SET ex.admissions (WHERE = (hosp EQ 3));
RUN;

The WHERE clause on the SET statement is used to choose those observations for which the hospital identifier corresponds to the VA. This is more efficient than a subsetting IF, though the result is the same.

SQL code:
PROC SQL FEEDBACK;
CREATE TABLE hosp3_adm2 AS
SELECT *
FROM ex.admissions
WHERE hosp EQ 3;
QUIT;

Here, the WHERE clause performs the same function as the subsetting IF above. Note that it is still part of the CREATE statement. A few additional features of SQL are demonstrated here in this simple query. First, the * is a “wild card” syntax, which essentially means “Select all the columns”. The FEEDBACK option on the PROC SQL statement requests an expansion of the query in the log. Useful in conjunction with the asterisk wild card, this results in the following statement in the SAS log:

NOTE: Statement transforms to:
select ADMISSIONS.PT_ID, ADMISSIONS.ADMDATE, ADMISSIONS.DISDATE, ADMISSIONS.MD, ADMISSIONS.HOSP, ADMISSIONS.DEST, ADMISSIONS.BP_SYS, ADMISSIONS.BP_DIA, ADMISSIONS.PRIMDX
from EX.ADMISSIONS
where ADMISSIONS.HOSP=3;

NOTE: Table WORK.HOSP3_ADM2 created, with 6 rows and 9 columns.

A subset of variables is shown in the output at the top of the next page.
Example 2: Selecting observations: Tarheel Hospital Admits

<table>
<thead>
<tr>
<th>pt_id</th>
<th>admdate</th>
<th>disdate</th>
<th>hosp</th>
</tr>
</thead>
<tbody>
<tr>
<td>003</td>
<td>17OCT2010</td>
<td>21OCT2010</td>
<td>03</td>
</tr>
<tr>
<td>003</td>
<td>15NOV2010</td>
<td>15NOV2010</td>
<td>03</td>
</tr>
<tr>
<td>005</td>
<td>11APR2010</td>
<td>28APR2010</td>
<td>03</td>
</tr>
<tr>
<td>008</td>
<td>01OCT2010</td>
<td>15OCT2010</td>
<td>03</td>
</tr>
<tr>
<td>008</td>
<td>26NOV2010</td>
<td>28NOV2010</td>
<td>03</td>
</tr>
<tr>
<td>014</td>
<td>17JAN2011</td>
<td>20JAN2011</td>
<td>03</td>
</tr>
<tr>
<td>018</td>
<td>01NOV2010</td>
<td>15NOV2010</td>
<td>03</td>
</tr>
<tr>
<td>018</td>
<td>26DEC2010</td>
<td>08JAN2011</td>
<td>03</td>
</tr>
</tbody>
</table>

Example 3: Creating a new variable

In this example we want to create a variable called DXGRP that categorizes the primary diagnosis into one of three categories (myocardial infarction [MI], congestive heart failure [CHF] or other), based on the ICD-9 code.

**DATA step code:**

```plaintext
DATA grouping ;
SET ex.admissions ;
LENGTH dxgrp $5 ;
IF primdx EQ: '410' THEN dxgrp = 'MI' ;
ELSE IF primdx EQ: '428' THEN dxgrp = 'CHF';
ELSE dxgrp = 'other' ;
RUN;
```

The useful EQ: comparison operator (or, equivalently =:) allows us to select all values of PRIMDX that begin with the specified string of characters, regardless of subsequent characters.

**SQL code:**

```plaintext
PROC SQL;
CREATE TABLE grouping2 AS
SELECT *,
    CASE
        WHEN primdx LIKE '410%' THEN 'MI'
        WHEN primdx LIKE '428%' THEN 'CHF'
        ELSE 'other'
    END AS dxgrp
FROM ex.admissions;
QUIT;
```

Here, the CASE clause of PROC SQL is used in conjunction with the LIKE keyword and the % wildcard to define the new variable DXGRP. Below is a partial listing of the results.

<table>
<thead>
<tr>
<th>pt_id</th>
<th>admdate</th>
<th>primdx</th>
<th>dxgrp</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>07FEB2010</td>
<td>410</td>
<td>MI</td>
</tr>
<tr>
<td>001</td>
<td>12APR2010</td>
<td>428.2</td>
<td>CHF</td>
</tr>
<tr>
<td>001</td>
<td>10SEP2010</td>
<td>813.9</td>
<td>other</td>
</tr>
<tr>
<td>001</td>
<td>19SEP2010</td>
<td>428.4</td>
<td>CHF</td>
</tr>
<tr>
<td>003</td>
<td>17OCT2010</td>
<td>410.01</td>
<td>MI</td>
</tr>
<tr>
<td>003</td>
<td>15NOV2010</td>
<td>431</td>
<td>other</td>
</tr>
<tr>
<td>004</td>
<td>18JUN2010</td>
<td>434.1</td>
<td>other</td>
</tr>
<tr>
<td>005</td>
<td>19JAN2010</td>
<td>411.81</td>
<td>other</td>
</tr>
<tr>
<td>005</td>
<td>10MAR2010</td>
<td>410.9</td>
<td>MI</td>
</tr>
<tr>
<td>007</td>
<td>28JUL2010</td>
<td>155</td>
<td>other</td>
</tr>
</tbody>
</table>
There is one tricky thing to remember about the CASE clause. It cannot be used to assign new values to a variable that already exists on the data set – it is only for creating “new” variables.

EXAMPLE 4: SELECTING ROWS BASED ON A CREATED VARIABLE

In this example we want to create a variable corresponding to the number of days of the hospital stay and select only those stays with duration of at least 14 days. Usually, both the admission date and discharge date are considered days of stay.

### DATA Step code:

```sas
DATA twowks1 ;
SET ex.admissions (KEEP = pt_id hosp admdate disdate) ;
ATTRIB los LENGTH=4 LABEL='Length of Stay';
los = (disdate - admdate) + 1;
IF los GE 14 ;
RUN;
```

**SQL code:**

```sql
PROC SQL;
CREATE TABLE twowks2 AS
SELECT pt_id, hosp, admdate, disdate, 
(disdate-admdate) + 1 AS los LENGTH=4 LABEL='Length of Stay'
FROM ex.admissions
WHERE CALCULATED los GE 14;
QUIT;
```

Here, we see the creation of a new column and the assignment of a column alias (LOS). Attributes can also be added; they could include a FORMAT as well. There is also one more subtle feature here: the CALCULATED keyword is required to indicate that the column LOS doesn’t exist on the input table (EX.ADMISSIONS) but is calculated during the query execution.

### Example 4: Selecting observations based on created variable

<table>
<thead>
<tr>
<th>pt_id</th>
<th>admdate</th>
<th>disdate</th>
<th>hosp</th>
<th>los</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>12APR2010</td>
<td>25APR2010</td>
<td>01</td>
<td>14</td>
</tr>
<tr>
<td>005</td>
<td>11APR2010</td>
<td>28APR2010</td>
<td>03</td>
<td>18</td>
</tr>
<tr>
<td>007</td>
<td>28JUL2010</td>
<td>10AUG2010</td>
<td>02</td>
<td>14</td>
</tr>
<tr>
<td>008</td>
<td>01OCT2010</td>
<td>15OCT2010</td>
<td>03</td>
<td>15</td>
</tr>
<tr>
<td>009</td>
<td>15DEC2010</td>
<td>04JAN2011</td>
<td>02</td>
<td>21</td>
</tr>
<tr>
<td>018</td>
<td>01NOV2010</td>
<td>15NOV2010</td>
<td>03</td>
<td>15</td>
</tr>
<tr>
<td>018</td>
<td>26DEC2010</td>
<td>08JAN2011</td>
<td>03</td>
<td>14</td>
</tr>
<tr>
<td>020</td>
<td>08OCT2011</td>
<td>01NOV2011</td>
<td>01</td>
<td>25</td>
</tr>
</tbody>
</table>

On the other hand, it is not required to assign an alias to a calculated column. The following would be perfectly valid and would select the same observations:

```sas
SELECT  pt_id, hosp, admdate, disdate, (disdate-admdate) + 1
FROM ex.admissions
WHERE (disdate - admdate) + 1 GE 14;
```

However, SAS will assign an arbitrary, system-dependent variable name to this column in the resulting table. If this column had a LABEL, it would print at the top of the column in the output, though the underlying variable name would still be the undecipherable _TEMA001 or something similar.

EXAMPLE 5: SELECTING ROWS IN ONE TABLE BASED ON INFORMATION FROM ANOTHER TABLE

Returning to the example (Example 2) of selecting admissions to the Tarheel hospital, let’s say we didn’t know which value of the HOSP variable corresponded to the Tarheel hospital. The information that provides a “cross-walk” between the hospital identifier code and the hospital name is in the HOSPITALS data set.
DATA Step Code:

PROC SORT DATA = ex.admissions OUT=admits;
BY hosp ;
RUN;
DATA tarheel_adm1 (DROP = hospname) ;  
MERGE admits (IN=adm)
  ex.hospitals (IN=va KEEP = hosp_id hospname
      RENAME = (hosp_id=hosp)
      WHERE = (hospname = 'Tarheel'));
    BY hosp ;
    IF adm AND va;
RUN;
PROC SORT;
BY pt_id admdate;
RUN;

We first need to sort the ADMISSIONS data set by the hospital code, and then merge it with the HOSPITALS data set, renaming the hospital code variable and selecting only those observations with a hospital name of “Tarheel”. If we want the admission to again be in ascending order by patient ID and admission date, another sort is required. The resulting data set is the same as in Example 2.

PROC SQL Code:

PROC SQL;
CREATE TABLE tarheel_adm AS
SELECT *
FROM ex.admissions
WHERE hosp IN
  (SELECT hosp_id
   FROM ex.hospitals
   WHERE hospname = 'Tarheel')
ORDER BY pt_id, admdate ;
QUIT;

This procedure contains an example of a subquery, or a query-expression that is nested within another query-expression. The value of the hospital identifier (HOSP) on the ADMISSIONS data set is compared to the result of a subquery of the HOSPITALS data set. Using IN (rather than EQ) in the WHERE clause allows for the possibility that the subquery might return more than a single value (i.e. if more than one hospital had a name of “Tarheel”). Note that no columns are added to the resulting table from the HOSPITALS data set, which is a point of distinction between a subquery and a join. No explicit sorting is required for this subquery to work. The ORDER BY clause dictates the sort order of the output data set. The output is identical to that shown for Example 2.

EXAMPLE 6: USING SUMMARY FUNCTIONS

Our next task is to count the number of admissions for each of the patients with at least one admission. We also want to calculate the minimum and maximum length of stay for each patient.

DATA Step Code:

DATA admsum1 ;
SET ex.admissions ;
BY pt_id;
** (1) Initialization;
IF FIRST.pt_id THEN DO;
nstays = 0;
minlos = .;
maxlos = .;
END;

Hands-On Workshops
NESUG 2012
We process the input data set by PT_ID. The DATA step has three sections. First, when the input observation is the first one for each subject, we initialize each of the summary variables. Next, in the accumulation phase we increment our counter and determine if the current stay is the longest or shortest for this patient. The RETAIN statement permits these comparisons. Finally, when it is the last input observation for a given PT_ID, we output an observation to our summary data set, keeping only the ID and the summary variables. If we kept any other variables, their values in the output data set would be the values they had for the last observation for each subject, and the output data set would still have one observation for each patient in the ADMISSIONS file (i.e. 15).

** PROC SQL code:**

```
PROC SQL;
CREATE TABLE admsum2 AS
    SELECT pt_id, COUNT(*) AS nstays,
           MIN(disdate - admdate + 1) AS minlos,
           MAX(disdate - admdate + 1) AS maxlos
    FROM ex.admissions
    GROUP BY pt_id ;
QUIT;
```

Two new features of PROC SQL are introduced here. First, the GROUP BY clause instructs SQL what the groupings are over which to perform any summary functions. Second, the summary functions include COUNT, which is the SQL name for the N or FREQ functions used in other SAS procedures. The COUNT(*) syntax essentially says count the rows in each GROUP BY group. The summary columns are each given an alias. A portion of the output is below.

<table>
<thead>
<tr>
<th>pt_id</th>
<th>nstays</th>
<th>minlos</th>
<th>maxlos</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>4</td>
<td>2</td>
<td>14</td>
</tr>
<tr>
<td>003</td>
<td>2</td>
<td>1</td>
<td>5</td>
</tr>
<tr>
<td>004</td>
<td>1</td>
<td>7</td>
<td>7</td>
</tr>
<tr>
<td>005</td>
<td>4</td>
<td>2</td>
<td>18</td>
</tr>
<tr>
<td>006</td>
<td>1</td>
<td>3</td>
<td>3</td>
</tr>
<tr>
<td>007</td>
<td>2</td>
<td>8</td>
<td>14</td>
</tr>
<tr>
<td>008</td>
<td>4</td>
<td>3</td>
<td>15</td>
</tr>
</tbody>
</table>

If we selected any columns other than the grouping column(s) and the summary variables, the resulting table would have a row for every row in the input table (i.e. 30) with the summary information duplicated across rows with a common value of the grouping variable (PT_ID), and we’d get messages similar to the following in the log:

```
NOTE: The query requires remerging summary statistics back with the original data.
NOTE: Table WORK.ADMSUM2 created, with 30 rows and 5 columns.
```

Sometimes this “re-merging” is useful (see Example 8 below), but it is not what we want for this situation. There is a NOREMERGE option that can ask SAS to generate an error statement if you attempt to do this. Consider turning this option on if you know that you do not intend a re-merge in your application.
EXAMPLE 7: SELECTION BASED ON SUMMARY FUNCTIONS

Let’s say we want to identify potential blood pressure outliers. We’d like to select all those observations that are two standard deviations or further from the mean. As usual, we start with the non-SQL method, starting at the top of the next page.

**DATA Step Code:**

```plaintext
PROC SUMMARY DATA= ex.admissions ;
   VAR bp_sys ;
   OUTPUT OUT=bpstats MEAN(bp_sys)= mean_sys STD(bp_sys) = sd_sys ;
RUN;
DATA hi_sys1 ;
   SET bpstats (keep=mean_sys sd_sys) ex.admissions ;
   IF _N_ EQ 1 THEN DO;
      high = mean_sys + 2*(sd_sys) ;
      low = mean_sys - 2*(sd_sys) ;
      DELETE;
   END;
   RETAIN high low ;
   IF (bp_sys GE high) OR (bp_sys LE low) ;
   DROP mean_sys sd_sys high low ;
RUN;
```

PROC SUMMARY generates the desired statistics. We then concatenate this one-observation data set (BPSTATS) with ADMISSIONS, RETAINing the high and low cutoffs to allow the necessary comparison to select potential outliers.

**PROC SQL Code:**

```plaintext
PROC SQL ;
   CREATE TABLE hi_sys2 AS
      SELECT * FROM ex.admissions
      WHERE (bp_sys GE (SELECT MEAN(bp_sys)+ 2*STD(bp_sys)
         FROM ex.admissions))
      OR (bp_sys LE (SELECT MEAN(bp_sys) - 2*STD(bp_sys)
         FROM ex.admissions));
QUIT;
```

In the SQL version, the summary functions are used in two similar subqueries of the same table to generate the values against which the systolic blood pressure for each observation in the outer query is compared. There is no GROUP BY clause because we are generating the summary values for the entire data set. The result (selected columns) is shown below.

**Example 7: Selection based on Summary Functions**

<table>
<thead>
<tr>
<th>Patient ID</th>
<th>Admit Date</th>
<th>Systolic</th>
<th>Diastolic</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>12APR2010</td>
<td>230</td>
<td>101</td>
</tr>
<tr>
<td>003</td>
<td>15NOV2010</td>
<td>74</td>
<td>40</td>
</tr>
<tr>
<td>009</td>
<td>15DEC2010</td>
<td>228</td>
<td>92</td>
</tr>
</tbody>
</table>
EXAMPLE 8: SELECTION BASED ON SUMMARY FUNCTION WITH “RE-MERGE”

This example adds a small twist to the prior one by requiring that we select admissions with extreme systolic blood pressure values within groups defined by discharge destination. The variable DEST is 1 for those who are discharged home, 2 for those discharged to a rehabilitation facility and 9 for those who die.

DATA Step Code:

```sas
PROC SUMMARY DATA= ex.admissions NWAY;
   CLASS dest ;
   VAR bp_sys ;
   OUTPUT OUT=bpstats2 MEAN(bp_sys)=mean_sys STD(bp_sys)=sd_sys ;
RUN;
PROC SORT DATA = ex.admissions OUT=admissions;
   BY dest ;
RUN;
DATA hi_sys3 ;
   MERGE admissions (KEEP = pt_id bp_sys bp_dia dest)
       bpstats2 (KEEP = dest mean_sys sd_sys);
   BY dest ;
   IF bp_sys GE mean_sys + 2*(sd_sys) OR
       bp_sys LE mean_sys - 2*(sd_sys) ;
FORMAT mean_sys sd_sys 6.2;
RUN;
```

To obtain the desired summary statistics for blood pressure we again use PROC SUMMARY. However, this time we include a CLASS statement to obtain the mean and standard deviation for each category of discharge destination and include the NWAY option so the BPSTATS2 data set does not include the overall statistics. The ADMISSIONS data set must be sorted by DEST before merging in the destination-specific means and standard deviations. A subsetting IF selects the desired observations.

PROC SQL Code:

```sas
PROC SQL;
   CREATE TABLE hi_sys4 AS
       SELECT pt_id, bp_sys, bp_dia, dest,
           MEAN(bp_sys) AS mean_sys FORMAT=6.2,
           STD(bp_sys) AS sd_sys FORMAT=6.2
       FROM ex.admissions
       GROUP BY dest
       HAVING bp_sys GE (mean_sys + 2*sd_sys)
           OR bp_sys LE (mean_sys - 2*sd_sys) ;
QUIT;
```

In some ways the SQL code for this example, in which the statistics are generated and the selection of rows are made separately for each BY group, is simpler than the last one where the process was done for the sample as a whole. This example doesn’t require a subquery. Rather it relies on a “re-merging” of the summary statistics for each GROUPing back with the ungrouped data, permitting the row-by-row comparisons needed to select the outliers. A new keyword is introduced here as well. HAVING acts on groups in a manner analogous to the way a WHERE clause operates on rows. A HAVING expression usually is proceeded by a GROUP BY clause, which defines the group that the HAVING expression evaluates, and the query must include one or more summary functions.

Example 8: Select using Summary Functions with re-merge

<table>
<thead>
<tr>
<th>PT_ID</th>
<th>BP_SYS</th>
<th>BP_DIA</th>
<th>DEST</th>
<th>MEAN_SYS</th>
<th>SD_SYS</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>230</td>
<td>101</td>
<td>1</td>
<td>162.62</td>
<td>29.04</td>
</tr>
<tr>
<td>018</td>
<td>199</td>
<td>93</td>
<td>2</td>
<td>149.31</td>
<td>20.36</td>
</tr>
</tbody>
</table>
EXAMPLE 9: IDENTIFYING DUPLICATES

This example demonstrates another use of a HAVING expression. We wish to select observations from the DOCTORS data set that are not unique with respect to the physician identifier. In other words we want to pull out all the records for the doctors who have admitting privileges at more than one hospital. We’d like them in order by the physician’s last name.

**DATA Step Code:**

```
DATA selmd1 ;
    SET ex.doctors (KEEP = md_id lastname hospadm
                     RENAME = (hospadm=hospital));
    BY md_id ;
    IF NOT (FIRST.md_id AND LAST.md_id) ;
RUN;
PROC SORT DATA=selmd1;  BY lastname hospital ;
RUN;
```

Processing BY MD_ID with this subsetting IF will produce the desired result. Note that if you just want to eliminate duplicates, you can use PROC SORT with the NODUPKEY option (and the DUPOUT= option will output the deleted records to a file, but I prefer the above method because it allows me to examine the sets of duplicate records – for example to determine if all should be maintained in the file. The SQL code below achieves the same result.

**PROC SQL Code:**

```
PROC SQL ;
    CREATE TABLE selmd2 AS
        SELECT md_id, lastname, hospadm AS hospital
        FROM ex.doctors
        GROUP BY md_id
        HAVING COUNT(*) GE 2
        ORDER BY lastname, hospital
    QUIT;
```

Applying the GROUP BY clause, the query first counts how many rows are associated with each doctor. The HAVING expression then selects the rows that meet the following condition: being part of a group having more than one row. The output is shown below.

**Example 9: Identifying duplicates**

<table>
<thead>
<tr>
<th>md_id</th>
<th>lastname</th>
<th>hospital</th>
</tr>
</thead>
<tbody>
<tr>
<td>7803</td>
<td>Avitable</td>
<td>02</td>
</tr>
<tr>
<td>7803</td>
<td>Avitable</td>
<td>03</td>
</tr>
<tr>
<td>1972</td>
<td>Fitzhugh</td>
<td>01</td>
</tr>
<tr>
<td>1972</td>
<td>Fitzhugh</td>
<td>02</td>
</tr>
<tr>
<td>3274</td>
<td>Hanratty</td>
<td>01</td>
</tr>
<tr>
<td>3274</td>
<td>Hanratty</td>
<td>02</td>
</tr>
<tr>
<td>3274</td>
<td>Hanratty</td>
<td>03</td>
</tr>
<tr>
<td>2322</td>
<td>MacArthur</td>
<td>01</td>
</tr>
<tr>
<td>2322</td>
<td>MacArthur</td>
<td>02</td>
</tr>
<tr>
<td>8081</td>
<td>Premnath</td>
<td>03</td>
</tr>
<tr>
<td>8081</td>
<td>Premnath</td>
<td>05</td>
</tr>
<tr>
<td>8081</td>
<td>Premnath</td>
<td>06</td>
</tr>
</tbody>
</table>
EXAMPLE 10: CREATION OF TWO DATA SETS FROM ONE

For the next example suppose we want to create separate data sets for the admissions in 2010 and 2011.

**DATA Step Code:**

```plaintext
DATA admit10 admit11 ;
SET ex.admissions ;
IF YEAR(admdate) = 2010 THEN OUTPUT admit10;
ELSE IF YEAR(admdate) = 2011 THEN OUTPUT admit11;
RUN;
```

This is readily accomplished by including two data set names in the DATA statement and directing observations to the appropriate output data set based on the value of the YEAR function.

**PROC SQL Code:**

```plaintext
PROC SQL ;
CREATE TABLE admit10 AS
   SELECT * FROM ex.admissions
   WHERE YEAR(admdate) = 2010;
CREATE TABLE admit11 AS
   SELECT * FROM ex.admissions
   WHERE YEAR(admdate) = 2011;
QUIT;
```

This invocation of PROC SQL includes two separate but nearly identical CREATE statements, one for each output table. Almost all of the functions that are available in the DATA step are also available in PROC SQL. A partial listing of each of the output tables is shown below.

<table>
<thead>
<tr>
<th>pt_id</th>
<th>admdate</th>
<th>md</th>
<th>hosp</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>07FEB2010</td>
<td>3274</td>
<td>01</td>
</tr>
<tr>
<td>001</td>
<td>12APR2010</td>
<td>1972</td>
<td>01</td>
</tr>
<tr>
<td>001</td>
<td>10SEP2010</td>
<td>3274</td>
<td>02</td>
</tr>
<tr>
<td>001</td>
<td>19SEP2010</td>
<td>3274</td>
<td>05</td>
</tr>
<tr>
<td>003</td>
<td>17OCT2010</td>
<td>8081</td>
<td>03</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>pt_id</th>
<th>admdate</th>
<th>md</th>
<th>hosp</th>
</tr>
</thead>
<tbody>
<tr>
<td>006</td>
<td>11SEP2011</td>
<td>8081</td>
<td>05</td>
</tr>
<tr>
<td>010</td>
<td>30NOV2011</td>
<td>2322</td>
<td>04</td>
</tr>
<tr>
<td>012</td>
<td>04JAN2011</td>
<td>4003</td>
<td>05</td>
</tr>
<tr>
<td>014</td>
<td>17JAN2011</td>
<td>7803</td>
<td>03</td>
</tr>
<tr>
<td>015</td>
<td>25MAY2011</td>
<td>4003</td>
<td>05</td>
</tr>
</tbody>
</table>

EXAMPLE 11: CONCATENATION

If we were starting with separate data sets for each year and wanted to combine into a single data set, ordered as in our original data set (by PT_ID and ADMDATE), the following code would serve.

**DATA Step Code:**

```plaintext
DATA alladm1 ;
SET admit10 admit11 ;
BY pt_id ;
RUN;
```
The BY statement is needed to ensure the desired ordering. Without it, all the 2010 admissions would precede all the 2011 admissions.

**PROC SQL Code:**

```sql
PROC SQL;
CREATE TABLE alladm2 AS
  SELECT * FROM admit10
  UNION ALL CORRESPONDING
  SELECT * FROM admit11
ORDER BY pt_id;
QUIT;
```

Generally the UNION set operator concatenates the component data sets so that the resulting table is in the same sort order as each of the original components. However, the ORDER BY clause ensures that this is the case. With these particular input data, the ALL and CORRESPONDING keywords are not necessary; that is, the result would be the same without them; however, this is not always the case. CORRESPONDING makes sure that SQL behaves in a DATA step like fashion with respect to column names. This means that it will store data for like-named columns from the two sources in a column with that name; otherwise, it simply lines columns up in the order they are in the data set. This is not a problem here because the columns are in the same order on the ADMIT10 and ADMIT11 data sets, but in some circumstances it would be critical to getting the desired result! The ALL keyword prevents SQL from eliminating duplicate rows from the output: eliminating duplicates is the default behavior for the UNION operator. This doesn’t mean rows that are duplicates on just PT_ID but rather on all columns that are being selected, which in this example is all the columns (thanks to SELECT *). There are no records that are identical coming from the two sources, so the ALL keyword has no effect. Nonetheless, these two options are needed to make the behavior most consistent with how the SET statement is operating in the DATA step. For further detail on the behavior of UNION and these options (and potential pitfalls of not using them!), see Williams (2012).

**EXAMPLE 12: SELECTING RECORDS UNIQUE TO ONE TABLE**

Continuing to work with our two single-year admissions data sets, suppose we wish to identify the patients that had admissions in 2010 but not 2011. Most likely one would use a MERGE to do this with the DATA Step:

**DATA Step Code:**

```data
DATA only2010 ;
MERGE admit10 (IN=in10 keep = pt_id)
  admit11 (IN=in11 keep = pt_id);
BY pt_id ;
IF in10 AND NOT in11 ;
IF FIRST.pt_id ;
RUN;
```

The boolean IN= variables allow us to identify the patients with records in the 2010 data set but not the 2011 data set. The subsetting IF statement (IF FIRST.pt_id) assures that we get just a single record for each patient – without this statement, for any patients that had multiple admissions in 2010 but none in 2011 (pt_id’s 005, 011, 008 and 018) there would be multiple rows.

**PROC SQL Code:**

```sql
PROC SQL;
CREATE TABLE only2010 AS
  SELECT pt_id FROM admit10
  EXCEPT
  SELECT pt_id FROM admit11;
QUIT;
```

The EXCEPT set operator identifies rows that are in the first operand (here ADMIT10) and not the second operand (ADMIT11). Note that there is no code included to eliminate duplicates. By default the EXCEPT operator excludes duplicates on the selected columns (just as we saw with the UNION operator in the previous example). If we wanted to include them we would add the keyword ALL after EXCEPT (i.e. “SELECT pt_id FROM admit10 EXCEPT ALL SELECT…”).
Example 12: Selecting Records Unique to One Table

<table>
<thead>
<tr>
<th>Obs</th>
<th>pt_id</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>001</td>
</tr>
<tr>
<td>2</td>
<td>003</td>
</tr>
<tr>
<td>3</td>
<td>004</td>
</tr>
<tr>
<td>4</td>
<td>005</td>
</tr>
<tr>
<td>5</td>
<td>007</td>
</tr>
<tr>
<td>6</td>
<td>008</td>
</tr>
<tr>
<td>7</td>
<td>009</td>
</tr>
<tr>
<td>8</td>
<td>018</td>
</tr>
</tbody>
</table>

Additionally, if we wanted to keep other variables but just eliminate PT_ID matches, we would have to move away from the EXCEPT operator and use a subquery like this:

**ALTERNATIVE PROC SQL Code:**

```
PROC SQL;
CREATE TABLE only20_A AS
SELECT *
FROM admit10
WHERE pt_id NOT IN
(SELECT pt_ID FROM admit11);
QUIT;
```

This would result in the selection of twenty observations – essentially all of the admissions in 2010 except those for PT_ID 012.

**EXAMPLE 13: INNER JOIN OF TWO TABLES**

A join combines data from two or more tables to produce a single result table; the table resulting from an inner join contains rows that have one or more matches in the other table(s).

**DATA Step Code:**

```
DATA admits1;
MERGE ex.admissions (IN=adm KEEP = pt_id admdat disdate hosp md)
   ex.patients  (IN=pts KEEP = id lastname sex primmd)
   RENAME = (id=pt_id);
BY pt_id;
IF adm AND pts;
RUN;
```

Selection based on the IN= temporary variables does the trick. Note that this produces the desired result partly because although there may be multiple admissions for each patient, the PATIENTS data set has only one observation for each value of the key variable PT_ID. The information on each record for a given PT_ID in the PATIENTS data set is replicated onto each observation in the output data set.

**PROC SQL code:**

```
PROC SQL;
CREATE TABLE admits2 AS
SELECT pt_id, admdat, disdate, hosp, md, lastname, sex, primmd
FROM ex.admissions AS a,
ex.patients AS b
WHERE a.pt_id = b.id
ORDER BY a.pt_id, admdat;
QUIT;
```

The table aliases A and B are used here to clarify which ID variables are coming from which data set. They are not required here because there are no columns being selected here that exist on both input data sets. Note that the AS keyword is not
required, but it emphasizes that an alias is being assigned. The code above might be more commonly used for a simple inner join, but the following syntax produces the same result.

```
PROC SQL;
  CREATE TABLE admits2 AS
    SELECT pt_id, admdate, disdate, hosp, md, lastname, sex, primmd
    FROM ex.admissions INNER JOIN
        ex.patients
    ON pt_id = id
  ORDER BY pt_id, admdate;
QUIT;
```

This is also an example of an "equijoin" because the selection criteria is equality of a column in one table with a column in the second table. SAS MERGEs are always equijoins. In the abbreviated output below, only a subset of the 30 selected rows and 8 columns are shown.

```
Example 13: Inner Join of two tables

<table>
<thead>
<tr>
<th>pt_id</th>
<th>pt_id</th>
<th>admdate</th>
<th>hosp</th>
<th>md</th>
<th>lastname</th>
<th>primmd</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>001</td>
<td>07FEB2010</td>
<td>01</td>
<td>3274</td>
<td>Williams</td>
<td>1972</td>
</tr>
<tr>
<td>001</td>
<td>001</td>
<td>12APR2010</td>
<td>01</td>
<td>1972</td>
<td>Williams</td>
<td>1972</td>
</tr>
<tr>
<td>001</td>
<td>001</td>
<td>10SEP2010</td>
<td>02</td>
<td>3274</td>
<td>Williams</td>
<td>1972</td>
</tr>
<tr>
<td>001</td>
<td>001</td>
<td>19SEP2010</td>
<td>05</td>
<td>3274</td>
<td>Williams</td>
<td>1972</td>
</tr>
<tr>
<td>003</td>
<td>003</td>
<td>17OCT2010</td>
<td>03</td>
<td>8081</td>
<td>Gillette</td>
<td>8081</td>
</tr>
<tr>
<td>003</td>
<td>003</td>
<td>15NOV2010</td>
<td>03</td>
<td>2322</td>
<td>Gillette</td>
<td>8081</td>
</tr>
<tr>
<td>004</td>
<td>004</td>
<td>18JUN2010</td>
<td>02</td>
<td>7803</td>
<td>Wallace</td>
<td>4003</td>
</tr>
<tr>
<td>005</td>
<td>005</td>
<td>19JAN2010</td>
<td>01</td>
<td>1972</td>
<td>Abbott</td>
<td>1972</td>
</tr>
<tr>
<td>005</td>
<td>005</td>
<td>10MAR2010</td>
<td>01</td>
<td>1972</td>
<td>Abbott</td>
<td>1972</td>
</tr>
<tr>
<td>005</td>
<td>005</td>
<td>10APR2010</td>
<td>02</td>
<td>1972</td>
<td>Abbott</td>
<td>1972</td>
</tr>
<tr>
<td>005</td>
<td>005</td>
<td>11APR2010</td>
<td>03</td>
<td>7803</td>
<td>Abbott</td>
<td>1972</td>
</tr>
</tbody>
</table>
```

EXAMPLE 14: JOIN OF THREE TABLES WITH ROW SELECTION

Let's make it a little trickier! Let's say we now wish to identify patients who died in the hospital (DEST = 9); we want our resulting data set to include their age at death and the number of beds in the hospital. This requires obtaining information from three of our tables, with differing key fields.

```
DATA died1 (RENAME = (disdate=dthdate)) ;
  MERGE ex.admissions (IN=dth KEEP = pt_id disdate hosp dest
    WHERE = (dest=9))
      ex.patients (IN=pts KEEP = id dob RENAME = (id=pt_id));
  BY pt_id ;
  IF dth AND pts ;
    agedth = FLOOR(((disdate-dob)/365.25) ;
  DROP dest dob ;
RUN;

PROC SORT DATA=died1;
  BY hosp;
RUN;

DATA died1b ;
  MERGE died1 (IN=dth RENAME=(hosp=hosp_id))
    ex.hospitals (IN=hsp KEEP=hosp_id beds);
  BY hosp_id ;
  IF dth AND hsp ;
  DROP hosp_id;
```
The DATA step version requires two DATA steps and two SORTs.

**PROC SQL code:**

```sql
PROC SQL;
CREATE TABLE died2 AS
SELECT pt_id,
beds,
disdate AS dthdate,
INT((disdate-dob)/365.25) AS agedth
FROM ex.admissions, ex.hospitals, ex.patients
WHERE (pt_id = id) AND (hosp = hosp_id) AND dest EQ 9
ORDER BY pt_id;
QUIT;
```

With PROC SQL, we can query the combination of the three tables in one step because there is no requirement of a single key that links all of the inputs. The output data set contains 4 rows and the desired variables.

**Example 14: Join of three tables with row selection**

<table>
<thead>
<tr>
<th>pt_id</th>
<th>beds</th>
<th>dthdate</th>
<th>agedth</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>475</td>
<td>22SEP2010</td>
<td>79</td>
</tr>
<tr>
<td>003</td>
<td>724</td>
<td>15NOV2010</td>
<td>83</td>
</tr>
<tr>
<td>009</td>
<td>1176</td>
<td>04JAN2011</td>
<td>83</td>
</tr>
<tr>
<td>012</td>
<td>475</td>
<td>09JAN2011</td>
<td>75</td>
</tr>
</tbody>
</table>

**EXAMPLE 15: LEFT OUTER JOIN**

A left outer join is an inner join of two or more tables that is augmented with rows from the "left" table that do not match with any rows in the "right" table(s). For this example we want to produce a table that has a row for each hospital with an indicator of whether there were any admissions at that hospital.

**DATA Step Code:**

```sas
PROC SORT DATA = ex.admissions (KEEP = hosp)
   OUT=admits (RENAME=(hosp=hosp_id)) NODUPKEY;
   BY hosp ;
RUN;
DATA hosps1 ;
   MERGE ex.hospitals (IN=hosp)
   admits (IN=adm);
   BY hosp_id ;
   IF hosp ;
   hasadmit = adm ;
RUN;
```

If the duplicates were not removed from the ADMISSIONS data set, the output data set would have multiple observations for each hospital; hence NODUPKEY is used on the SORT. The temporary boolean IN= variable is made permanent in the MERGE step to create our indicator of having at least one record in the ADMISSIONS data set.

**PROC SQL code:**

```sql
PROC SQL;
CREATE TABLE hosps2 AS
SELECT DISTINCT a.*,
hosp IS NOT NULL AS hasadmit
```
FROM ex.hospitals a LEFT JOIN ex.admissions b  
ON a.hosp_id = b.hosp ;
QUIT;

The keyword DISTINCT causes SQL to eliminate duplicate rows from the resulting table. The expression “hosp IS NOT NULL AS hasadmit” assigns the alias HASADMIT to a new column whose value is TRUE (i.e. 1) if a given HOSP_ID from the HOSPITALS table has a matching HOSP value in the ADMISSIONS table. The resulting data set – shown below – has one row for each hospital and a binary indicator of admissions as desired.

Example 15: Left Outer Join

<table>
<thead>
<tr>
<th>hosp_id</th>
<th>hospname</th>
<th>zip</th>
<th>beds</th>
<th>has_er</th>
<th>hasadmit</th>
</tr>
</thead>
<tbody>
<tr>
<td>01</td>
<td>Deacon</td>
<td>27105</td>
<td>202</td>
<td>Y</td>
<td>1</td>
</tr>
<tr>
<td>02</td>
<td>City</td>
<td>27607</td>
<td>1176</td>
<td>Y</td>
<td>1</td>
</tr>
<tr>
<td>03</td>
<td>Tarheel</td>
<td>27514</td>
<td>724</td>
<td>Y</td>
<td>1</td>
</tr>
<tr>
<td>04</td>
<td>Peace</td>
<td>28585</td>
<td>839</td>
<td>N</td>
<td>1</td>
</tr>
<tr>
<td>05</td>
<td>BlueDevil</td>
<td>27708</td>
<td>475</td>
<td>Y</td>
<td>1</td>
</tr>
<tr>
<td>06</td>
<td>Wolfpack</td>
<td>27603</td>
<td>650</td>
<td>N</td>
<td>1</td>
</tr>
<tr>
<td>07</td>
<td>FarOut</td>
<td>27850</td>
<td>68</td>
<td>Y</td>
<td>0</td>
</tr>
</tbody>
</table>

EXAMPLE 16: INNER JOIN OF TWO TABLES AND A QUERY

One of the items combined in a join can itself be a query. In this case we want to identify and select the admissions for which patients were treated by their primary physicians (PRIMMD on the PATIENTS data set). We want the output data set to include the doctor’s name and the patient’s name.

**DATA Step Code:**

```sql
DATA primdoc (DROP = primmd);
  MERGE ex.admissions (IN=adm KEEP = pt_id admdate disdate hosp md)
    ex.patients (IN=pts KEEP = id lastname primmd RENAME=(id=pt_id));
    BY pt_id ;
    IF adm AND pts AND (md EQ primmd) ;
RUN;
PROC SORT DATA=primdoc; BY md; RUN;

DATA doctors ;
  SET ex.doctors (KEEP = md_id lastname);
  BY md_id ;
  IF FIRST.md_id ;
RUN;

DATA primdocla ;
  MERGE primdoc (IN=p RENAME=(lastname=ptname md=md_id))
    doctors (RENAME = (lastname=mdname));
    BY md_id ;
    IF p ;
RUN;

PROC SORT DATA=primdocla ;
BY pt_id admdate; RUN;
```

The first DATA step above selects the admissions for which patients saw their primary physicians. The second DATA step eliminates duplicate records for the same physician. If this were not done, the final MERGE would be a many-to-many merge and would not produce the desired result. (Note that we could also have done a SORT with NODUPKEY). The final DATA step (with MERGE) simply adds the physician name to the selected admissions. Both LASTNAME variables are RENAMED to prevent the physician name from overwriting the patient name. Here is the SQL version:

**PROC SQL Code:**

```sql
PROC SQL ;
```
CREATE TABLE primdoc2 AS
  SELECT pt_id, admdate, disdate, hosp, md_id,
    b.lastname AS ptname,
    c.lastname AS mdname
  FROM ex.admissions a, ex.patients b,
  (SELECT DISTINCT md_id, lastname
    FROM ex.doctors) c
WHERE (a.pt_id EQ b.id) AND
  (a.md EQ b.primmd) AND
  (a.md EQ c.md_id)
ORDER BY a.pt_id, admdate ;
QUIT;

The third “table” listed in the FROM clause is itself a query that selects non-duplicate physician ID's and names from the DOCTORS data set. The result of this query can be aliased just like a table, and here the aliases (b and c) are required so that the two LASTNAME columns can be distinguished. The ultimate row selection is very straightforward. Sometimes for a complicated query like this it is helpful to break it down into separate queries. Results are shown below:

Example 16: Inner Join of Two Tables and a Query

<table>
<thead>
<tr>
<th>pt_id</th>
<th>admdate</th>
<th>disdate</th>
<th>md_id</th>
<th>hosp</th>
<th>ptname</th>
<th>mdname</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>12APR2010</td>
<td>25APR2010</td>
<td>1972</td>
<td>01</td>
<td>Williams</td>
<td>Fitzhugh</td>
</tr>
<tr>
<td>003</td>
<td>17OCT2010</td>
<td>21OCT2010</td>
<td>8081</td>
<td>03</td>
<td>Gillette</td>
<td>Premnath</td>
</tr>
<tr>
<td>005</td>
<td>19JAN2010</td>
<td>22JAN2010</td>
<td>1972</td>
<td>01</td>
<td>Abbott</td>
<td>Fitzhugh</td>
</tr>
<tr>
<td>005</td>
<td>10MAR2010</td>
<td>18MAR2010</td>
<td>1972</td>
<td>01</td>
<td>Abbott</td>
<td>Fitzhugh</td>
</tr>
<tr>
<td>005</td>
<td>10APR2010</td>
<td>11APR2010</td>
<td>1972</td>
<td>02</td>
<td>Abbott</td>
<td>Fitzhugh</td>
</tr>
<tr>
<td>007</td>
<td>28JUL2010</td>
<td>10AUG2010</td>
<td>3274</td>
<td>02</td>
<td>Pedersen</td>
<td>Hanratty</td>
</tr>
<tr>
<td>010</td>
<td>30NOV2011</td>
<td>06DEC2011</td>
<td>2322</td>
<td>04</td>
<td>Alberts</td>
<td>MacArthur</td>
</tr>
<tr>
<td>018</td>
<td>01NOV2010</td>
<td>15NOV2010</td>
<td>1972</td>
<td>03</td>
<td>Baker</td>
<td>Fitzhugh</td>
</tr>
<tr>
<td>018</td>
<td>26DEC2010</td>
<td>08JAN2011</td>
<td>1972</td>
<td>03</td>
<td>Baker</td>
<td>Fitzhugh</td>
</tr>
</tbody>
</table>

CONCLUSION

I hope that the examples presented in this paper have convinced you that PROC SQL is an extremely versatile tool for the manipulation of data sets. Row selection, summarization, the combination of information from multiple input sources, and the ordering of the output can often be achieved in a single statement! Another compelling reason for becoming comfortable with SQL is that many information systems store data in foreign databases, such as ORACLE, Microsoft Access or SQL Server. If these data are to be manipulated and analyzed in SAS, frequently PROC SQL provides the link (e.g. through ODBC). Perhaps seeing some familiar DATA step techniques followed by a call to the SQL procedure that achieves the same result will give you the impetus to try SQL or dig into it a bit more deeply. I’ll close with three observations that I hope will provide some encouragement.

First, it is always useful to have many different techniques to draw on when tackling a challenging data management task. And using a new project or assignment as an opportunity to learn some new methods makes you a more valuable employee – and probably a more fulfilled one as well.

Second, on the technical side, the most complicated nested query can usually be broken down into manageable parts – start from the “inside” (the most nested expressions) and work your way out. While it may be possible to do it all in one statement, you don’t have to. Try making each level of nesting into a separate SELECT statement, using aliases liberally, with a final statement that connects the results of these simpler statements. Once this is working, you can start building the parts back together again – if you wish.

Finally, in constructing these examples, I was struck that using SQL forces one to think about data sets in a slightly different way, focusing more on the relationships among tables than the structure of any one table. In fact, it makes one realize that a database is defined not only by the component tables but just as importantly by the linkages among them. This broadened perspective can provide insight into building better databases as well as writing better programs to access and manipulate them.

ACKNOWLEDGMENTS

Many thanks to Peter Charpentier, Evelyne Gahbauer and Virginia Towle for their careful reading of and extremely helpful comments on earlier versions of this paper.
SAS is a registered trademark or trademark of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are registered trademarks or trademarks of their respective companies.

REFERENCES AND RECOMMENDED READING

CONTACT INFORMATION
I welcome your comments or questions.

Christianna S. Williams, PhD
Email: Christianna.S.Williams@gmail.com
APPENDIX

Full listing of data sets for the Examples

<table>
<thead>
<tr>
<th>pt_id</th>
<th>admdate</th>
<th>disdate</th>
<th>md</th>
<th>hosp</th>
<th>dest</th>
<th>bp_sys</th>
<th>bp_dia</th>
<th>primdx</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>07FEB2010</td>
<td>08FEB2010</td>
<td>3274</td>
<td>01</td>
<td>1</td>
<td>188</td>
<td>85</td>
<td>410</td>
</tr>
<tr>
<td>001</td>
<td>12APR2010</td>
<td>25APR2010</td>
<td>1972</td>
<td>01</td>
<td>1</td>
<td>230</td>
<td>101</td>
<td>428.2</td>
</tr>
<tr>
<td>001</td>
<td>10SEP2010</td>
<td>19SEP2010</td>
<td>3274</td>
<td>02</td>
<td>2</td>
<td>170</td>
<td>78</td>
<td>813.9</td>
</tr>
<tr>
<td>001</td>
<td>19SEP2010</td>
<td>22SEP2010</td>
<td>3274</td>
<td>05</td>
<td>9</td>
<td>185</td>
<td>94</td>
<td>428.4</td>
</tr>
<tr>
<td>003</td>
<td>17OCT2010</td>
<td>21OCT2010</td>
<td>8081</td>
<td>03</td>
<td>1</td>
<td>155</td>
<td>92</td>
<td>410.01</td>
</tr>
<tr>
<td>003</td>
<td>15NOV2010</td>
<td>15NOV2010</td>
<td>2322</td>
<td>03</td>
<td>9</td>
<td>74</td>
<td>40</td>
<td>431</td>
</tr>
<tr>
<td>004</td>
<td>18JUN2010</td>
<td>24JUN2010</td>
<td>7803</td>
<td>02</td>
<td>2</td>
<td>140</td>
<td>78</td>
<td>434.1</td>
</tr>
<tr>
<td>005</td>
<td>19JAN2010</td>
<td>22JAN2010</td>
<td>1972</td>
<td>01</td>
<td>1</td>
<td>148</td>
<td>84</td>
<td>411.81</td>
</tr>
<tr>
<td>005</td>
<td>10MAR2010</td>
<td>18MAR2010</td>
<td>1972</td>
<td>01</td>
<td>1</td>
<td>160</td>
<td>90</td>
<td>410.9</td>
</tr>
<tr>
<td>005</td>
<td>10APR2010</td>
<td>11APR2010</td>
<td>1972</td>
<td>02</td>
<td>2</td>
<td>150</td>
<td>89</td>
<td>411</td>
</tr>
<tr>
<td>005</td>
<td>11APR2010</td>
<td>28APR2010</td>
<td>7803</td>
<td>03</td>
<td>1</td>
<td>145</td>
<td>91</td>
<td>411</td>
</tr>
<tr>
<td>006</td>
<td>11SEP2011</td>
<td>13SEP2011</td>
<td>8081</td>
<td>05</td>
<td>2</td>
<td>129</td>
<td>83</td>
<td>820.01</td>
</tr>
<tr>
<td>007</td>
<td>28JUL2010</td>
<td>10AUG2010</td>
<td>3274</td>
<td>02</td>
<td>2</td>
<td>136</td>
<td>72</td>
<td>155</td>
</tr>
<tr>
<td>007</td>
<td>08SEP2010</td>
<td>15SEP2010</td>
<td>8081</td>
<td>02</td>
<td>2</td>
<td>138</td>
<td>71</td>
<td>155</td>
</tr>
<tr>
<td>008</td>
<td>13APR2010</td>
<td>19APR2010</td>
<td>1972</td>
<td>02</td>
<td>1</td>
<td>140</td>
<td>80</td>
<td>428.4</td>
</tr>
<tr>
<td>008</td>
<td>01OCT2010</td>
<td>15OCT2010</td>
<td>3274</td>
<td>03</td>
<td>1</td>
<td>145</td>
<td>74</td>
<td>820.8</td>
</tr>
<tr>
<td>008</td>
<td>26NOV2010</td>
<td>28NOV2010</td>
<td>2322</td>
<td>03</td>
<td>2</td>
<td>135</td>
<td>76</td>
<td>V54.8</td>
</tr>
<tr>
<td>008</td>
<td>10DEC2010</td>
<td>12DEC2010</td>
<td>8081</td>
<td>06</td>
<td>2</td>
<td>132</td>
<td>78</td>
<td>V54.8</td>
</tr>
<tr>
<td>009</td>
<td>15DEC2010</td>
<td>04JAN2011</td>
<td>1972</td>
<td>02</td>
<td>9</td>
<td>228</td>
<td>92</td>
<td>410.1</td>
</tr>
<tr>
<td>010</td>
<td>30NOV2011</td>
<td>06DEC2011</td>
<td>2322</td>
<td>04</td>
<td>1</td>
<td>147</td>
<td>84</td>
<td>E886.3</td>
</tr>
<tr>
<td>012</td>
<td>12AUG2010</td>
<td>16AUG2010</td>
<td>4003</td>
<td>05</td>
<td>1</td>
<td>187</td>
<td>106</td>
<td>410.52</td>
</tr>
<tr>
<td>012</td>
<td>04JAN2011</td>
<td>09JAN2011</td>
<td>4003</td>
<td>05</td>
<td>9</td>
<td>201</td>
<td>98</td>
<td>433.4</td>
</tr>
<tr>
<td>014</td>
<td>17JAN2011</td>
<td>20JAN2011</td>
<td>7803</td>
<td>03</td>
<td>1</td>
<td>162</td>
<td>93</td>
<td>414.1</td>
</tr>
<tr>
<td>015</td>
<td>25MAY2011</td>
<td>06JUN2011</td>
<td>4003</td>
<td>05</td>
<td>2</td>
<td>142</td>
<td>81</td>
<td>820.8</td>
</tr>
<tr>
<td>015</td>
<td>17AUG2011</td>
<td>24AUG2011</td>
<td>4003</td>
<td>05</td>
<td>2</td>
<td>138</td>
<td>79</td>
<td>38.2</td>
</tr>
<tr>
<td>016</td>
<td>25JUL2011</td>
<td>30JUL2011</td>
<td>7803</td>
<td>02</td>
<td>1</td>
<td>189</td>
<td>101</td>
<td>412.1</td>
</tr>
<tr>
<td>018</td>
<td>15NOV2010</td>
<td>15NOV2010</td>
<td>1972</td>
<td>03</td>
<td>2</td>
<td>170</td>
<td>88</td>
<td>428.1</td>
</tr>
<tr>
<td>018</td>
<td>26DEC2010</td>
<td>08JAN2011</td>
<td>1972</td>
<td>03</td>
<td>2</td>
<td>199</td>
<td>93</td>
<td>428.1</td>
</tr>
<tr>
<td>020</td>
<td>08OCT2011</td>
<td>01NOV2011</td>
<td>2322</td>
<td>01</td>
<td>2</td>
<td>162</td>
<td>99</td>
<td>434</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>id</th>
<th>sex</th>
<th>primmd</th>
<th>lastname</th>
<th>firstname</th>
<th>dob</th>
<th>zipcode</th>
</tr>
</thead>
<tbody>
<tr>
<td>001</td>
<td>1</td>
<td>1972</td>
<td>Williams</td>
<td>Hugh</td>
<td>10AUG1931</td>
<td>27516</td>
</tr>
<tr>
<td>002</td>
<td>2</td>
<td>1972</td>
<td>Franklin</td>
<td>Susan</td>
<td>17MAR1938</td>
<td>27402</td>
</tr>
<tr>
<td>003</td>
<td>1</td>
<td>8081</td>
<td>Gillette</td>
<td>Michael</td>
<td>02JUL1927</td>
<td>29401</td>
</tr>
<tr>
<td>004</td>
<td>1</td>
<td>4003</td>
<td>Wallace</td>
<td>Geoffrey</td>
<td>25MAY1925</td>
<td>27699</td>
</tr>
<tr>
<td>005</td>
<td>2</td>
<td>1972</td>
<td>Abbott</td>
<td>Celeste</td>
<td>31AUG1940</td>
<td>27114</td>
</tr>
<tr>
<td>006</td>
<td>1</td>
<td>2322</td>
<td>Capel</td>
<td>Anthony</td>
<td>12APR1929</td>
<td>27155</td>
</tr>
<tr>
<td>007</td>
<td>1</td>
<td>3274</td>
<td>Pedersen</td>
<td>Lars</td>
<td>07FEB1938</td>
<td>27516</td>
</tr>
<tr>
<td>008</td>
<td>2</td>
<td>4003</td>
<td>Lieberman</td>
<td>Marianne</td>
<td>09NOV1942</td>
<td>27604</td>
</tr>
<tr>
<td>009</td>
<td>2</td>
<td>3274</td>
<td>Jacobson</td>
<td>Frances</td>
<td>15SEP1927</td>
<td>27708</td>
</tr>
<tr>
<td>010</td>
<td>2</td>
<td>2322</td>
<td>Alberts</td>
<td>Josephine</td>
<td>14OCT1948</td>
<td>28544</td>
</tr>
<tr>
<td>011</td>
<td>2</td>
<td>1972</td>
<td>Erickson</td>
<td>Karen</td>
<td>04NOV1926</td>
<td>29904</td>
</tr>
<tr>
<td>012</td>
<td>1</td>
<td>7803</td>
<td>Collins</td>
<td>Albert</td>
<td>16JUN1935</td>
<td>27340</td>
</tr>
<tr>
<td>013</td>
<td>1</td>
<td>4003</td>
<td>Greene</td>
<td>Riley</td>
<td>03AUG1946</td>
<td>27615</td>
</tr>
<tr>
<td>014</td>
<td>1</td>
<td>8034</td>
<td>Dohiman</td>
<td>Henrik</td>
<td>09NOV1950</td>
<td>27516</td>
</tr>
<tr>
<td>015</td>
<td>2</td>
<td>3274</td>
<td>Zakur</td>
<td>Hannah</td>
<td>06JUN1937</td>
<td>28117</td>
</tr>
<tr>
<td>016</td>
<td>1</td>
<td>1972</td>
<td>DeLucia</td>
<td>Antonio</td>
<td>17JUN1923</td>
<td>28083</td>
</tr>
<tr>
<td>017</td>
<td>1</td>
<td>2322</td>
<td>Cohen</td>
<td>Adam</td>
<td>17APR1935</td>
<td>27511</td>
</tr>
<tr>
<td>018</td>
<td>1</td>
<td>1972</td>
<td>Baker</td>
<td>Shelby</td>
<td>13FEB1940</td>
<td>27533</td>
</tr>
<tr>
<td>019</td>
<td>2</td>
<td>4003</td>
<td>Wallace</td>
<td>Judith</td>
<td>01FEB1933</td>
<td>26083</td>
</tr>
<tr>
<td>020</td>
<td>2</td>
<td>7803</td>
<td>Carrier</td>
<td>Sarah</td>
<td>07AUG1935</td>
<td>28357</td>
</tr>
</tbody>
</table>
### EX.HOSPITALS (N=7)

<table>
<thead>
<tr>
<th>hosp_id</th>
<th>hospname</th>
<th>zip</th>
<th>beds</th>
<th>has_er</th>
</tr>
</thead>
<tbody>
<tr>
<td>01</td>
<td>Deacon</td>
<td>27105</td>
<td>202</td>
<td>Y</td>
</tr>
<tr>
<td>02</td>
<td>City</td>
<td>27607</td>
<td>1176</td>
<td>Y</td>
</tr>
<tr>
<td>03</td>
<td>Tarheel</td>
<td>27514</td>
<td>724</td>
<td>Y</td>
</tr>
<tr>
<td>04</td>
<td>Peace</td>
<td>28585</td>
<td>839</td>
<td>N</td>
</tr>
<tr>
<td>05</td>
<td>BlueDevil</td>
<td>27708</td>
<td>475</td>
<td>Y</td>
</tr>
<tr>
<td>06</td>
<td>Wolfpack</td>
<td>27603</td>
<td>650</td>
<td>N</td>
</tr>
<tr>
<td>07</td>
<td>FarOut</td>
<td>27850</td>
<td>68</td>
<td>Y</td>
</tr>
</tbody>
</table>

### EX.DOCTORS (N=14)

<table>
<thead>
<tr>
<th>md_id</th>
<th>lastname</th>
<th>zip_office</th>
<th>hospadm</th>
</tr>
</thead>
<tbody>
<tr>
<td>1972</td>
<td>Fitzhugh</td>
<td>27105</td>
<td>01</td>
</tr>
<tr>
<td>1972</td>
<td>Fitzhugh</td>
<td>27105</td>
<td>02</td>
</tr>
<tr>
<td>2322</td>
<td>MacArthur</td>
<td>27514</td>
<td>01</td>
</tr>
<tr>
<td>2322</td>
<td>MacArthur</td>
<td>27514</td>
<td>03</td>
</tr>
<tr>
<td>2998</td>
<td>Rosenberg</td>
<td>28585</td>
<td>04</td>
</tr>
<tr>
<td>3274</td>
<td>Hanratty</td>
<td>27708</td>
<td>01</td>
</tr>
<tr>
<td>3274</td>
<td>Hanratty</td>
<td>27708</td>
<td>02</td>
</tr>
<tr>
<td>3274</td>
<td>Hanratty</td>
<td>27708</td>
<td>03</td>
</tr>
<tr>
<td>4003</td>
<td>Colantonio</td>
<td>28544</td>
<td>05</td>
</tr>
<tr>
<td>7803</td>
<td>Avitable</td>
<td>27105</td>
<td>02</td>
</tr>
<tr>
<td>7803</td>
<td>Avitable</td>
<td>27105</td>
<td>03</td>
</tr>
<tr>
<td>8081</td>
<td>Premnath</td>
<td>27607</td>
<td>02</td>
</tr>
<tr>
<td>8081</td>
<td>Premnath</td>
<td>27607</td>
<td>05</td>
</tr>
<tr>
<td>8081</td>
<td>Premnath</td>
<td>27607</td>
<td>06</td>
</tr>
</tbody>
</table>